Elliptic integrals

1 Arc length of an ellipse (elliptic integral of the second kind)

Find the arc length of an ellipse

(%i1) `declare(a, constant);`

(%o1) done

(%i2) `declare(b,constant);`

(%o2) done

(%i3) `assume(a>b);`

(%o3) $[b < a]$

(%i4) `assume(xi > 0);`

(%o4) $[xi > 0]$

We parametrize the equation of an ellipse

(%i5) `x(ϕ) := a·sin(ϕ);`

(%o5) $x(\phi) := a \sin(\phi)$

(%i6) `y(ϕ):= b·cos(ϕ);`

(%o6) $y(\phi) := b \cos(\phi)$

Here we compute

(%i7) `('ds^2 = 'dx^2 + 'dy^2)/'dϕ^2;`

(%o7) $$\frac{ds^2}{d\phi^2} = \frac{dy^2 + dx^2}{d\phi^2}$$

(%i8) `expr1: expand((diff(x(ϕ))^2 + diff(y(ϕ))^2)/(del(ϕ)^2));`

expr1 $b^2 \sin(\phi)^2 + a^2 \cos(\phi)^2$

The arc length is integral (ds) = integral(ds/dϕ * dϕ)

(%i9) `int1: integrate(sqrt(expr1),ϕ,0,xi);`

int1 $$\int_0^{xi} \sqrt{b^2 \sin(\phi)^2 + a^2 \cos(\phi)^2}\, d\phi$$

We manipulate the integrand to put it in a canonical form

(%i10) `dpart(int1,1,1);`

(%o10) $$\int_0^{xi} \sqrt{\boxed{b^2 \sin(\phi)^2 + a^2 \cos(\phi)^2}}\, d\phi$$

(%i11) `s1: trigsimp(part(int1,1,1)+a^2) -a^2;`

s1 $(b^2 - a^2) \sin(\phi)^2 + a^2$

(%i12) `s2: (part(s1,1)/a^2) + part(s1,2)/a^2;`

s2 $$\frac{(b^2 - a^2) \sin(\phi)^2}{a^2} + 1$$

(%i13) `int2: substpart(s2,int1,1,1);`

int2 $$\int_0^{xi} \sqrt{\frac{(b^2 - a^2) \sin(\phi)^2}{a^2} + 1}\, d\phi$$

Let m = (a^2 - b^2)/a^2

This is the definition of the elliptic integral.

(%i14) `elliptic_e_def: ratsubst(m,(a^2-b^2)/a^2,int2);`

elliptic_e_def $$\int_0^{xi} \sqrt{1 - m \sin(\phi)^2}\, d\phi$$

(%i15) elliptic_e_artisanal(xi,m) := integrate(sqrt(1−m·sin(t)^2),t,0,xi);

(%o15) $\text{elliptic\_e\_artisanal}(xi, m) := \int_0^{xi} \sqrt{1 - m \sin(t)^2}\, dt$

Test: if we set m = 0, we obtain the integral of 1, evaluated from 0 to xi:

(%i16) elliptic_e_artisanal(xi,0);

(%o16) xi

(%i17) elliptic_e(xi,0);

(%o17) xi

Plot of the integrand: the elliptic function is then the area under that curve from 0 to a given number xi. We see that the function should or t.

(%i18) wxplot2d([sqrt(1−0.5·sin(t)^2)], [t,−%pi,6·%pi],
[gnuplot_postamble, "set zeroaxis;"])$

(%t18)



Here is the plot of the elliptic function: indeed, it increases monotonically.

(%i19) wxplot2d([elliptic_e(x,0.9)], [x,−%pi,6·%pi],
[gnuplot_postamble, "set zeroaxis;"])$

(%t19)



2 Elliptic integral of the first kind

(%i20) `assume(a>0);`

(%o20) $[0 < a]$

In the solution of the problem for the period of a simple pendulum, we encounter the integral

(%i21) `periodIntegral: integrate(1/sqrt(cos(theta)−cos(a)),theta,0,a);`

periodIntegral $\int_0^a \frac{1}{\sqrt{\cos(\mathrm{theta}) - \cos(a)}} d\,\mathrm{theta}$

This integral motivates the definition of the elliptic functions of the first kind, but after several changes of variable.

(%i22) `integrand1: part(periodIntegral,1);`

integrand1 $\frac{1}{\sqrt{\cos(\mathrm{theta}) - \cos(a)}}$

(%i23) `integrand2: ratsubst(2·sin(theta/2)^2,1−cos(theta),integrand1);`

integrand2 $\frac{1}{\sqrt{-\left(2\sin\left(\frac{\mathrm{theta}}{2}\right)^2\right) - \cos(a) + 1}}$

(%i24) `integrand3: ratsubst(2·sin(a/2)^2, 1−cos(a), integrand2);`

integrand3 $\frac{1}{\sqrt{2\sin\left(\frac{a}{2}\right)^2 - 2\sin\left(\frac{\mathrm{theta}}{2}\right)^2}}$

(%i25) `integrand4: ratsubst(b,sin(a/2),integrand3);`

integrand4 $\frac{1}{\sqrt{2\ b^2 - 2\ \sin\left(\frac{\theta}{2}\right)^2}}$

(%i26) integrand5: ratsubst(sin(phi),sin(theta/2)/b, integrand4);

integrand5 $\frac{1}{|b|\sqrt{2 - 2\ \sin(\phi)^2}}$

(%i27) integrand6: trigsimp(integrand5);

integrand6 $\frac{1}{\sqrt{2}\ |b|\ |\cos(\phi)|}$

(%i28) /*differential */
eq1: diff(sin(phi) = sin(theta/2)/b);

eq1 $\cos(\phi)\ \mathrm{del}(\phi) = \frac{\cos\left(\frac{\theta}{2}\right)\ \mathrm{del}(\theta)}{2\ b}$

(%i29) eq2: ratsubst(sqrt(1 − sin(theta/2)^2),cos(theta/2), eq1);

eq2 $\cos(\phi)\ \mathrm{del}(\phi) = \frac{\sqrt{1 - \sin\left(\frac{\theta}{2}\right)^2}\ \mathrm{del}(\theta)}{2\ b}$

(%i30) eq3: ratsubst(sin(phi),sin(theta/2)/b,eq2);

eq3 $\cos(\phi)\ \mathrm{del}(\phi) = \frac{\sqrt{1 - b^2\ \sin(\phi)^2}\ \mathrm{del}(\theta)}{2\ b}$

(%i31) solve(eq3,del(theta));

(%o31) $\left[\mathrm{del}(\theta) = \frac{2\ b\ \cos(\phi)\ \mathrm{del}(\phi)}{\sqrt{1 - b^2\ \sin(\phi)^2}}\right]$

(%i32) integrand7: ratsimp(integrand6·2·b·cos(phi)/sqrt(1−b^2·sin(phi)^2));

integrand7 $\frac{\sqrt{2}\ b\ \cos(\phi)}{|b|\sqrt{1 - b^2\ \sin(\phi)^2}\ |\cos(\phi)|}$

(%i33) assume(cos(phi)>0, b>0);

(%o33) $[\cos(\phi) > 0, 0 < b]$

(%i34) expand(integrand7);

(%o34) $\frac{\sqrt{2}}{\sqrt{1 - b^2\ \sin(\phi)^2}}$

Finally, we arrive at the elliptic integral of the first kind (multiplied by the factor of sqrt(2)).
We transform the integration limits with the substitutions made and obtain

(%i37) /* New integral: elliptic function of the first kind * sqrt(2) */
integrate(expand(integrand7),phi,0,%pi/2);

(%o37) $\sqrt{2}\int_0^{\frac{\pi}{2}} \frac{1}{\sqrt{1 - b^2\ \sin(\phi)^2}}\, d\phi$

For small oscillations of a pendulum, it is interesting to develop the integral into a series

(%i41) series1: taylor(1/sqrt(1−w),w,0,1);

series1 $1 + \frac{w}{2} + \ldots$

(%i45) approxintegrand1: expand(ratsubst(b^2·sin(phi)^2,w,expand(series1)));

approxintegrand1 $\frac{b^2\ \sin(\phi)^2}{2} + 1$

(%i47) approx1: expand(integrate(approxintegrand1, phi, 0, %pi/2));

approx1 $\frac{\pi b^2}{8}+\frac{\pi}{2}$

which is, returning to the original variable

(%i50) approx2: expand(ratsubst(sin(a/2),b,approx1));

approx2 $\frac{\pi \sin\left(\frac{a}{2}\right)^2}{8}+\frac{\pi}{2}$

or for sin(a/2) ~ a/2

(%i52) expand(ratsubst(a/2,sin(a/2),approx2));

(%o52) $\frac{\pi a^2}{32}+\frac{\pi}{2}$

2.1 Plots of the elliptic function of the second kind

Integrand

```
wxplot2d([1/sqrt(1-0.7·sin(x)^2)], [x,-%pi,6·%pi],
 [gnuplot_postamble, "set zeroaxis;"])$
```

(%t24)



Function

```
wxplot2d([elliptic_f(x,0.7)], [x,-%pi,6·%pi],
 [gnuplot_postamble, "set zeroaxis;"])$
```

(%t25)



25
20
15
10
5
0
-5
0
5
10
15
x
elliptic_f(x,0.7)